
3 (f) Samples McStas: 

virtual experiments and sample environment
E. Farhi <farhi@ill.eu> - Thursday Sept 19th 2013.

References:

Refer to the Workshops and Conferences McStas page at <http://mcstas.org/workshop/>

Scope:

In this exercise, we shall make use of a powder sample in a diffractometer set-up. A diffractogram will 
be simulated and effects of the sample size on the resolution will be estimated. Then, a sample 
environment will be added to estimate its contribution in a measurement. Last, the same methodology 
will be suggested with single crystal and liquid samples.

3F.1: a constant wavelength diffractometer with a powder sample

Open McGUI and click on the Edit button.
1. In the Editor, select the menu item Edit/Insert/Instrument Template
2. Save your instrument as Exercise3f_samples.instr
3. Add input parameters DEFINE INSTRUMENT Exercise3f_samples(lambda=2, string 

sample="Na2Ca3Al2F14.laz")
4. In the TRACE after Origin, insert a Source_simple(radius=0.005, 

dist=4,xwidth=0.02,yheight=0.02,lambda0=lambda,dlambda=0.01*lambda
, focus_xw=0.01, focus_yh=0.01). To insert a component, you may position the cursor 
at the insertion location, and use the Insert menu to select the component you wish to include. 
This is a 1% monochromatic source.

5. At 4 m, add a sample PowderN(reflections=sample, 
radius=0.005,yheight=0.02,d_phi=50)

6. Around the sample, add a banana detector Monitor_nD(radius=2,yheight=1, 
options="banana theta y, auto", bins=180)

7. Click on the McGUI 'Run' button, then 'Start' to launch the simulation. 
8. Observe onion rings (scattering from a powder gives rings which angle give the atomic 

spacing). Press 'L' key for log-scale
9. Save this instrument (menu File/Save)
10. To avoid seeing the direct beam, you may customize the behaviour of the sample component by 

removing all non scattered neutrons. For this, add an EXTEND block after its 'AT' keyword 
line. Re-run the simulation.

EXTEND %{ 
  if (!SCATTERED) ABSORB; 
%}

A list of predefined sample files is available from the Help/Library Index item and then [data files] (top 
bar).

Figure 3F.1: a constant wavelength diffraction pattern

3F.2: a tof version of the diffractometer

We do not have time to build a full chopper system to select incoming neutrons. Instead, we select the 
wavelength range at the source.

Save the instrument as Exercise3f_samples_ESS.instr

Change the source component into an ESS_moderator_long with parameters: 
    Lmin = lambda*0.95, Lmax = lambda*1.05, focus_xw = 0.01,
    focus_yh = 0.01, cold_frac = 0.01, n_pulses = 1, dist=4

Change the Monitor into an angle-tof one, with options="banana angle time, auto" 

Click Run again. Compare with the constant wavelength version.

Figure 3F.2: tof version of the Powder sample on an ESS thermal source

3F.3 Adding a sample environment

With McStas, any concentric geometry should be described symmetrically w.r.t. the 
sample position, e.g. :

COMPONENT entry_side= Comp(blah, concentric=1)

COMPONENT sample= …

COMPONENT exit_side= COPY(entry_side)(concentric=0)

This works for the powder (PowderN), single crystal (Single_crystal) and liquid/amorphous/glassy 
(Isotropic_Sqw) materials.

Goal: surround the previous sample with a cylinder of Aluminium

1. Get back the initial Exercise3f_samples.instr description (not the ESS version)
2. Before the Sample, add a cylinder entry_side=PowderN(reflections=”Al.laz”, 

radius=0.035, radius_i=0.035-0.0002,d_phi=50,tfrac=0.8,concentric=1) 
centred on the sample

3. After the sample, add a  exit_side=COPY(entry_side)(concentric=0)
4. Re-run simulation. Are there additional rings ?
5. In the DECLARE %{ … %} add int flag_env,flag_sample;
6. EXTEND the Origin with flag_env=flag_sample=0; 
7. EXTEND PowderN components with e.g. if (SCATTERED) flag_blah=1; where 

blah is 'sample' or 'env'
8. Duplicate the Detector with copies that only activate WHEN (flag_blah) 
9. Re-run. Compare the intensity from the sample and the environment.

You should get something like

Figure 3F.3: adding sample environment

Solution to 3F.1
/*******************************************************************************
*         McStas instrument definition URL=http://www.mcstas.org
*
* Instrument: Exercise3f_samples
*
* %Identification
* Written by: Luke Skywalker
* Date: Current Date
* Origin: The Moon
* Release: McStas
* Version: 0.2
* %INSTRUMENT_SITE: Death_Star
*
* Exercise3f_samples: first instrument
*
* %Description
* a diffractometer with a powder sample
*
* Example: mcrun Exercise3f_samples lambda=2
*
* %Parameters
* lambda: incident wavelength [Angs]
* sample: sample description file [string, laz/lau]
*
* %End
*******************************************************************************/

DEFINE INSTRUMENT Exercise3f_samples(lambda=2, string sample="Na2Ca3Al2F14.laz")

TRACE

COMPONENT Origin = Progress_bar()
  AT (0,0,0) ABSOLUTE

COMPONENT Source = Source_simple(
radius=0.005, dist=4,xwidth=0.02,yheight=0.02,lambda0=lambda,dlambda=0.01*lambda,
focus_xw=0.01, focus_yh=0.01)

  AT(0,0,0) RELATIVE PREVIOUS

COMPONENT  powder = PowderN(
reflections=sample, radius=0.005,yheight=0.02,d_phi=50)

  AT(0,0,4) RELATIVE PREVIOUS

COMPONENT monitor = Monitor_nD(
radius=2,yheight=1, options="banana theta y, auto", bins=180)

  AT(0,0,0) RELATIVE PREVIOUS

END

